Lógica Proposicional en Isabelle/HOL

Sofía Santiago Fernández

actualizado el 24 de noviembre de 2019

Resumen

Añadir introducción

Índice

1 Sintaxis

notation *insert* (- ▷ - [*56*,*55*] *55*)

En este apartado vamos a desarrollar los elementos de la sintaxis junto con varios resultados sobre los mismos. La lógica proposicional cuenta, fundamentalmente con:

Alfabeto: Se trata de una lista infinita de variables proposicionales, consideradas como expresiones cuya estructura interna no vamos a analizar.

Conectivas: Conectores que interactúan con los elementos del alfabeto. Pueden ser monarias que afectan a un único elemento o binarias que afectan a dos. En el primer grupo se encuentra le negación ($\neg$) y en el segundo vamos a considerar la conjunción ($\wedge$), la disyunción ($\vee$) y la implicación ($\longrightarrow$).

De este modo, informalmente diremos que una fórmula es el resultado de relacionar los elementos del alfabeto mediante las conectivas definidas anteriormente. En primer lugar, podemos definir las fórmulas atómicas como el tipo de fórmulas más básico, formadas únicamente por una variable porposicional del alfabeto. Por abuso de notación llamaremos *átomos* a las variables proposicionales del alfabeto. Aunque son intuitivamente equivalentes a las fórmulas atómicas, debemos recalcar su diferencia, pues los átomos son los elementos del alfabeto y las fórmulas atómicas son construcciones básicas de ellos. Este apunte es fundamental para entender el tipo correspondiente de fórmulas en nuestro programa.

En Isabelle, las fórmulas son entendidas como un árbol con distintos tipos de nodos, que son las conectivas, y hojas, que serán las fórmulas atómicas. De este modo, se define el tipo de las fórmulas como sigue.

datatype (*atoms*: *'a*) *formula* =
  Atom 'a
| *Bot* (⊥)
| *Not 'a formula* (¬)
| *And 'a formula 'a formula* (**infix** ∧ *68*)
| *Or 'a formula 'a formula* (**infix** ∨ *68*)
| *Imp 'a formula 'a formula* (**infixr** → *68*)

Como podemos observar en la definición, *formula* es un tipo de datos recursivo que se entiende como un árbol que relaciona elementos de un tipo *'a* cualquiera del alfabeto proposicional. En ella aparecen los siguientes elementos:

Constructores :

1. *Atom* :: *'a formula*
2. *Not* :: *'a* ⇒ *'a formula* ⇒ *'a formula*
3. *And* :: *'a* ⇒ *'a formula* ⇒ *'a formula* ⇒ *'a formula*
4. *Or* :: *'a* ⇒ *'a formula* ⇒ *'a formula* ⇒ *'a formula*
5. *Imp* :: *'a* ⇒ *'a formula* ⇒ *'a formula* ⇒ *'a formula*
6. *Bot* :: *'a formula*

Función de conjunto : *atoms* :: *'a formula* ⇒ *'a set*

Podemos observar que se define simultáneamente la función *atoms* de modo que al aplicarla sobre una fórmula nos devuelve el conjunto de los átomos que la componen. En particular, *Atom 'a* es la construcción de las fórmulas atómicas. *Bot* es un constructor que para cada tipo *'a* cualquiera construye la fórmula falsa cuyo símbolo queda retratado en la definición.

Observemos que para emplear *Bot* en solitario es necesario especificar el tipo $'a$.

value$(Bot{::}nat\ formula)$

Veamos ejemplos de obtención del conjunto de las variables proposicionales de las fórmulas mediante la función *atoms*.

value $atoms\ (Atom\ p) = \{p\}$

value $atoms\ (Not\ (Atom\ p)) = \{p\}$

En particular, al aplicar *atoms* sobre la construcción *Bot* nos devuelve el conjunto vacío, pues como hemos señalado, no contiene ninguna variable del alfabeto.

lemma $atoms\ Bot = \{\}$

lemma $atoms\ (Or\ (Atom\ p)\ Bot) = \{p\}$

El resto de elementos que aparecen son equivalentes a las conectivas binarias y la negación. Cabe señalar que el término *infix* que precede al símbolo de notación de los nodos nos señala que son infijos, y *infixr* se trata de un infijo asociado a la derecha. A continuación vamos a incluir el ejemplo de fórmula: $(p \longrightarrow q) \vee r$ y su árbol de formación correspondiente.

value $Or\ (Imp\ (Atom\ p)\ (Atom\ q))\ (Atom\ r)$

(Aquí debería salir el árbol pero no sé hacerlo)

Por otro lado, veamos cómo actúa la función *atoms* sobre fórmulas construidas con conectivas binarias, señalando los casos en que interactúan variables distintas y repetidas. Como se observa, por definición de conjunto, no contiene elementos repetidos.

lemma $atoms\ (Or\ (Imp\ (Atom\ p)\ (Atom\ q))\ (Atom\ r)) = \{p,q,r\}$

lemma $atoms\ (Or\ (Imp\ (Atom\ r)\ (Atom\ p))\ (Atom\ r)) = \{p,r\}$

En esta sección, para demostrar los distintos resultados utilizaremos la táctica *induction*, que hace uso del esquema de indución. Para el tipo de las fórmulas, el esquema inductivo se aplicará en cada uno de los casos de los constructores, desglosándose así seis casos correspondientes a las distintas conectivas, fórmula atómica y *Bot*. Además, todas las demostraciones sobre casos de conectivas binarias son equivalentes en esta sección, pues la construcción sintáctica de fórmulas es idéntica entre ellas. Estas se diferencian esencialmente en la connotación semántica, que veremos más adelante. Por tanto, para simplificar algunas demostraciones sintácticas más extensas, expondremos la prueba estructurada únicamente para uno de los casos de conectivas binarias.

Llegamos así al primer resultado de este apartado:

Lema 1.1 *Los átomos de cualquier fórmula conforman un conjunto finito.*

En Isabelle, se traduce del siguiente modo.

lemma *atoms-finite*[*simp*,*intro*!]: *finite* (*atoms F*)
  oops

El lema anterior contiene la notación *simp*,*intro*! a continuación del título para incluir este resultado en las tácticas automática (mediante *intro*!) y de simplificación (mediante *simp*). Esto ocurrirá en varios resultados de esta sección.

Por otro lado, el enunciado contiene la defición *finite S*, perteneciente a la teoría FiniteSet.thy. Se trata de una definición inductiva que nos permite la demostración del lema por simplificacion pues dentro de ella, las reglas que especifica se han añadido como tácticas de *simp* e *intro*!.

inductive finite :: *'a set* $\Rightarrow$ *bool*
where
emptyI [*simp*, *intro*!]: *finite* {}
| *insertI* [*simp*, *intro*!]: *finite A* $\Longrightarrow$ *finite* (*insert a A*)

Veamos la prueba detallada del resultado que, aunque se resulve fácilmente por simplificación, nos muestra un ejemplo claro de la estructura inductiva que nos acompañará en las siguientes demostraciones.

lemma *atoms-finite-detallada*: *finite* (*atoms F*)
proof (*induction F*)
case (*Atom x*)
then show *?case* **by** *simp*
next
case *Bot*
  then show *?case* **by** *simp*
next

case (*Not F*)
then show *?case* **by** *simp*
next
case (*And F1 F2*)
then show *?case* **by** *simp*
next
case (*Or F1 F2*)
then show *?case* **by** *simp*
next
case (*Imp F1 F2*)
then show *?case* **by** *simp*
qed

Las demostraciones aplicativa y automática son las siguientes respectivamente.

lemma *atoms-finite-aplicativa*: *finite* (*atoms F*)
apply (*induction F*)
apply *simp-all*
done

lemma *atoms-finite*[*simp*,*intro*!]: *finite* (*atoms F*)
by (*induction F*; *simp*)

1.1 Subfórmulas

Otra construcción natural a partir de la definición de fórmulas son las subfórmulas.

Definición 1.2 *La lista de las subfórmulas de una fórmula F (Subf(F)) se define recursivamente por:*

1. $[\bot]$ *si* F *es* $\bot$.
2. $[F]$ *si* F *es atómica.*
3. $F\#Subf(G)$ *si* F *es* $\neg G$.
4. $F\#Subf(G)@Subf(H)$ *si* F *es* $G{*}H$ *donde* $*$ *es cualquier conectiva binaria.*

En la definición anterior, # es el operador que añade un elemento al comienzo de una lista y @ concatena varias listas. Análogamente, vamos a definir la función primitiva recursiva *subformulae*, que nos dará una lista de todas las subfórmulas de una fórmula original obtenidas recursivamente.

primrec $subformulae :: 'a\ formula \Rightarrow 'a\ formula\ list$ **where**
$subformulae\ \bot = [\bot]\ |$
$subformulae\ (Atom\ k) = [Atom\ k]\ |$
$subformulae\ (Not\ F) = Not\ F\ \#\ subformulae\ F\ |$
$subformulae\ (And\ F\ G) = And\ F\ G\ \#\ subformulae\ F\ @\ subformulae\ G\ |$
$subformulae\ (Imp\ F\ G) = Imp\ F\ G\ \#\ subformulae\ F\ @\ subformulae\ G\ |$
$subformulae\ (Or\ F\ G) = Or\ F\ G\ \#\ subformulae\ F\ @\ subformulae\ G$

Siguiendo con los ejemplos, observemos cómo actúa *subformulae* en las distintas fórmulas.

value $subformulae\ (Atom\ p) = [Atom\ p]$

value $subformulae\ (Not\ (Atom\ p)) = [Not\ (Atom\ p),\ Atom\ p]$

value $subformulae\ (Or\ (Imp\ (Atom\ p)\ (Atom\ q))\ (Atom\ r)) =$
$[(Atom\ p \longrightarrow Atom\ q) \vee Atom\ r,\ Atom\ p \longrightarrow Atom\ q,\ Atom\ p,\ Atom\ q,\ Atom\ r]$

value $subformulae\ (And\ (Atom\ p)\ Bot) =$
$[And\ (Atom\ p)\ Bot,\ Atom\ p,\ Bot]$

En particular, al tratarse de una lista pueden aparecer elementos repetidos como se muestra a continuación.

value $subformulae\ (Or\ (Atom\ p)\ (Atom\ p)) =$
$[Or\ (Atom\ p)\ (Atom\ p),\ Atom\ p,\ Atom\ p]$

value $subformulae\ (Or\ (Atom\ p)\ (Atom\ p)) =$
$[Or\ (Atom\ p)\ (Atom\ p),\ Atom\ p] = False$

A partir de esta definición, aparecen varios resultados sencillos que demostraremos siguiendo tácticas similares a las empleadas en el lema anterior. Para ello, trabajaremos con conjuntos en vez de listas, pues poseen ventajas como la eliminación de elementos repetidos o las operaciones propias. De este modo, definimos *setSubformulae*, que convierte en un tipo conjunto la lista de subfórmulas anterior.

abbreviation $setSubformulae :: 'a\ formula \Rightarrow 'a\ formula\ set$ **where**
$setSubformulae\ F \equiv set\ (subformulae\ F)$

Debemos señalar que la elección de *abbreviation* para definirlo no es arbitraria, pues en caso contrario no permite la inclusión en las tácticas de simplificación de los lemas en los que actúa. Este hecho queda explicado en ... .

Por otra parte, observemos la diferencia de repetición con el ejemplo anterior.

value $setSubformulae\ (Or\ (Atom\ p)\ (Atom\ p)) = \{Or\ (Atom\ p)\ (Atom\ p),\ Atom\ p\}$

lemma $setSubformulae\ (Or\ (Imp\ (Atom\ p)\ (Atom\ q))\ (Atom\ r)) =$
$\{(Atom\ p \longrightarrow Atom\ q) \vee Atom\ r,\ Atom\ p \longrightarrow Atom\ q,\ Atom\ p,\ Atom\ q,\ Atom\ r\}$

Como hemos señalado, utilizaremos varios resultados de la teoría de conjuntos definida en Isabelle como Set.thy. Voy a especificar el esquema de las usadas en esta sección que no están incluidas en las tácticas de simplificación para aclarar las demostraciones detalladas que presentaré en este apartado.

$$\frac{A \subseteq B \wedge B \subseteq C}{A \subseteq C} \qquad (subset\text{-}trans)$$

$$\frac{c \in A \wedge A \subseteq B}{c \in B} \qquad (rev\text{-}subsetD)$$

$$\frac{A \subseteq C \wedge B \subseteq D}{A \cup B \subseteq C \cup D} \qquad (Un\text{-}mono)$$

$$A \subseteq A \cup B \qquad (Un\text{-}upper1)$$

$$B \subseteq A \cup B \qquad (Un\text{-}upper2)$$

Además, definiré alguna propiedad más que no aparece en la teoría de Isabelle y que utilizaremos con frecuencia. Su demostración será la automática.

lemma $subContUnionRev1$: $A \cup B \subseteq C \Longrightarrow A \subseteq C$

lemma *subContUnionRev2*: $A \cup B \subseteq C \Longrightarrow B \subseteq C$

Una vez aclarada la nueva función de conjuntos, vamos a demostrar el siguiente lema sirviéndonos de ella.

Lema 1.3 *El conjunto de los átomos de una fórmula está contenido en el conjunto de subfórmulas de la misma, es decir, las fórmulas atómicas son subfórmulas.*

En Isabelle, se especifica como sigue.

lemma *atoms-are-subformulae*: *Atom* ‘ *atoms F* $\subseteq$ *setSubformulae F*
oops

Este resultado es especialmente interesante para aclarar la naturaleza de la función *atoms* aplicada a una fórmula. De este modo, *Atom* ‘ *atoms F* se encarga de construir las fórmulas atómicas a partir de cada uno de los elementos del conjunto de átomos de la fórmula F, creando un conjunto de fórmulas atómicas. Para ello emplea el infijo ‘ definido como notación abreviada de (‘) que calcula la imagen de un conjunto en la teoría Set.thy.

$$f \text{ ‘ } A = \{y \mid \exists\, x{\in}A.\ y = f\, x\} \qquad (image\text{-}def)$$

Para aclarar su funcionamiento, veamos ejemplos para distintos casos de fórmulas.

value *Atom* ‘*atoms* (*Or* (*Atom p*) *Bot*) = {*Atom p*}

lemma *Atom* ‘*atoms* (*Or* (*Imp* (*Atom p*) (*Atom q*)) (*Atom r*)) = {*Atom p*, *Atom q*, *Atom r*}

lemma *Atom* ‘*atoms* (*Or* (*Imp* (*Atom p*) (*Atom p*)) (*Atom r*)) = {*Atom p*, *Atom r*}

Además, esta función tiene la siguiente propiedad sobre la unión de conjuntos que utilizaré en la demostración.

$$f \text{ ‘ } (A \cup B) = f \text{ ‘ } A \cup f \text{ ‘ } B \qquad (image\text{-}Un)$$

Una vez hecha la aclaración anterior, comencemos la demostración estructurada. Esta seguirá el esquema inductivo señalado con anterioridad. Debido a la extensión de la prueba demostraré de manera detallada únicamente uno de los casos de conectivas binarias. El resto son totalmente equivalentes y los dejaré indicados de manera automática. Además, en esta primera prueba voy a demostrar dcon detalle los casos básicos de *Atom x* y $\perp$, señalando las reglas y definiciones usadas. Sin embargo, podrían demostrarse de manera directa únicamente con simplificación.

lemma *atoms-are-subformulae-detallada*: *Atom* ‘ *atoms F* $\subseteq$ *setSubformulae F*
proof (*induction F*)
  case (*Atom x*)
  have *Atom* ‘ *atoms* (*Atom x*) = *Atom* ‘ {*x*} **by** *simp*
  also have ... = {*Atom x*} **by** *simp*
  also have ... $\subseteq$ {*Atom x*} **by** *simp*
  also have ... = *setSubformulae* (*Atom x*) **by** *simp*
  finally show *?case* **by** *simp*
next
  case *Bot*
  have *Atom* ‘ *atoms* (*Bot*) = *Atom* ‘ {} **by** *simp*
  also have ... = {} **by** *simp*
  also have ... $\subseteq$ {} **by** *simp*
  finally show *?case* **by** *simp*
next
  case (*Not F*)
  assume *H*:*Atom* ‘ *atoms F* $\subseteq$ *setSubformulae F*
  show *Atom* ‘ *atoms* (*Not F*) $\subseteq$ *setSubformulae* (*Not F*)
  proof −
    have {*Not F*} $\cup$ *setSubformulae F*= *setSubformulae* (*Not F*) **by** *simp*
   have *setSubformulae F* $\subseteq$ {*Not F*} $\cup$ *setSubformulae F* **by** (*rule Un-upper2*)
    then have *1*:*setSubformulae F* $\subseteq$ *setSubformulae* (*Not F*) **by** *simp*
    have *Atom* ‘ *atoms F* = *Atom* ‘ *atoms* (*Not F*) **by** *simp*
   then have *Atom* ‘ *atoms* (*Not F*) $\subseteq$ *setSubformulae F* **using** *H* **by** *simp*

    then show *?case* **using** *1* **by** (*rule subset-trans*)
  qed
next
  case (*And F1 F2*)
  assume *H1*:*Atom* ‘ *atoms F1* $\subseteq$ *setSubformulae F1*
  assume *H2*:*Atom* ‘ *atoms F2* $\subseteq$ *setSubformulae F2*

show *Atom ‘ atoms* (*And F1 F2*) $\subseteq$ *setSubformulae* (*And F1 F2*)
proof −
have *2*:(*Atom ‘ atoms F1*) $\cup$ (*Atom ‘ atoms F2*) $\subseteq$ *setSubformulae F1* $\cup$ *setSubformulae F2*
using *H1 H2* **by** (*rule Un-mono*)
have *setSubformulae* (*And F1 F2*) = {*And F1 F2*} $\cup$ (*setSubformulae F1* $\cup$ *setSubformulae F2*)
by *simp*
have *setSubformulae F1* $\cup$ *setSubformulae F2* $\subseteq$
{*And F1 F2*} $\cup$ (*setSubformulae F1* $\cup$ *setSubformulae F2*) **by** (*rule Un-upper2*)
then have *3*:*setSubformulae F1* $\cup$ *setSubformulae F2* $\subseteq$ *setSubformulae* (*And F1 F2*)
by *simp*
then have *setSubformulae F1* $\subseteq$ *setSubformulae* (*And F1 F2*) **by** *simp*
have *setSubformulae F2* $\subseteq$ *setSubformulae* (*And F1 F2*) **using** *3* **by** *simp*
also have *atoms* (*And F1 F2*) = *atoms F1* $\cup$ *atoms F2* **by** *simp*
then have *Atom ‘ atoms* (*And F1 F2*) = *Atom ‘* (*atoms F1* $\cup$ *atoms F2*) **by** *simp*
also have ... = *Atom ‘ atoms F1* $\cup$ *Atom ‘ atoms F2* **by** (*rule image-Un*)
then have *Atom ‘ atoms* (*And F1 F2*) = *Atom ‘ atoms F1* $\cup$ *Atom ‘ atoms F2* **by** *simp*
then have *Atom ‘ atoms* (*And F1 F2*) $\subseteq$ *setSubformulae F1* $\cup$ *setSubformulae F2* **using** *2* **by** *simp*
then show *Atom ‘ atoms* (*And F1 F2*) $\subseteq$ *setSubformulae* (*And F1 F2*) **using** *3*
by (*rule subset-trans*)
qed
next
case (*Or F1 F2*)
then show *?case* **by** *auto*
next
case (*Imp F1 F2*)
then show *?case* **by** *auto*
qed

Mostremos también la demostración automática con la definición original.

lemma *atoms-are-subformulae*: *Atom ‘ atoms F* $\subseteq$ *setSubformulae F*
by (*induction F*) *auto*

Otro resultado de esta sección es la propia pertenencia de una fórmula en el conjunto de sus subfórmulas.

Lema 1.4 *La propia fórmula pertence al conjunto de sus subfórmulas, es decir:* $F \in \mathit{SubfSet}(F)$.

Para que la propiedad esté bien definida para conjuntos, considero $\mathit{SubfSet}(\cdot)$ equivalente a *setSubformulae*, tal que al aplicarlo a la lista $\mathit{Subf}(F)$ nos devuelva el conjunto de sus subfórmulas.

A continuación incluimos el enunciado del lema con su demostración automática. Las demostraciones detallada y aplicativa son análogas a las del primer lema de la sección, utilizando únicamente inducción y simplificación. Para facilitar pruebas posteriores, vamos a añadir la regla a las tácticas de simplificación.

lemma *subformulae-self*[*simp*,*intro*]: $F \in$ *setSubformulae* F
  by (*induction* F) *simp-all*

La siguiente propiedad declara que el conjunto de átomos de una subfórmula está contenido en el conjunto de átomos de la propia fórmula.

Lema 1.5 *Sea* $G \in set(\mathit{Subf}(F))$, *entonces el conjunto de los átomos de* G *está contenido en los de* F.

Traducido a Isabelle:

lemma *subformula-atoms*: $G \in$ *setSubformulae* $F \Longrightarrow$ *atoms* $G \subseteq$ *atoms* F
  oops

Veamos su demostración estructurada. Desarrollaré uno de los casos de conectivas binarias, pues los demás son equivalentes.

lemma *subformula-atoms-estructurada-s*: $G \in$ *setSubformulae* $F \Longrightarrow$ *atoms* $G \subseteq$ *atoms* F
proof (*induction* F)
  case (*Atom* x)
  then show *?case* **by** *simp*
next
  case *Bot*
  then show *?case* **by** *simp*
next
  case (*Not* F)

assume *H1*: $G \in$ *setSubformulae* (*Not F*)
assume *H2*: $G \in$ *setSubformulae F* $\Longrightarrow$ *atoms G* $\subseteq$ *atoms F*
show *atoms G* $\subseteq$ *atoms* (*Not F*)
proof (*cases G = Not F*)
case *True*
then have *G = Not F* **by** *simp*
then show *atoms G* $\subseteq$ *atoms* (*Not F*) **by** *simp*
next
case *False*
then have *1*:$G \neq$ *Not F* **by** *simp*
have *setSubformulae* (*Not F*) = {*Not F*} $\cup$ *setSubformulae F* **by** *simp*
then have *2*:$G \in$ *setSubformulae F* **using** *1 H1* **by** *simp*
have *atoms F* = *atoms* (*Not F*) **by** *simp*
then show *atoms G* $\subseteq$ *atoms* (*Not F*) **using** *2 H2* **by** *simp*
qed
next
case (*And F1 F2*)
then show *?case* **by** *auto*
next
case (*Or F1 F2*)
assume *H3*: $G \in$ *setSubformulae* (*Or F1 F2*)
assume *H4*: $G \in$ *setSubformulae F1* $\Longrightarrow$ *atoms G* $\subseteq$ *atoms F1*
assume *H5*: $G \in$ *setSubformulae F2* $\Longrightarrow$ *atoms G* $\subseteq$ *atoms F2*
then show *atoms G* $\subseteq$ *atoms* (*Or F1 F2*)
proof (*cases G = Or F1 F2*)
case *True*
then have *G = Or F1 F2* **by** *simp*
then show *atoms G* $\subseteq$ *atoms* (*Or F1 F2*) **by** *simp*
next
case *False*
then have *3*:$G \neq$ *Or F1 F2* **by** *simp*
have *setSubformulae* (*Or F1 F2*) = {*Or F1 F2*} $\cup$ *setSubformulae F1* $\cup$ *setSubformulae F2* **by** *simp*
then have *4*:$G \in$ *setSubformulae F1* $\cup$ *setSubformulae F2* **using** *3 H3* **by** *simp*
have *atoms* (*Or F1 F2*) = *atoms F1* $\cup$ *atoms F2* **by** *simp*
then show *atoms G* $\subseteq$ *atoms* (*Or F1 F2*)
proof (*cases* $G \in$ *setSubformulae F1*)
case *True*
then have $G \in$ *setSubformulae F1* **by** *simp*
then have *5*:*atoms G* $\subseteq$ *atoms F1* **using** *H4* **by** *simp*

have *atoms F1* $\subseteq$ *atoms F1* $\cup$ *atoms F2* **by** (*rule Un-upper1*)
then have *6*:*atoms F1* $\subseteq$ *atoms* (*Or F1 F2*) **by** *simp*
show *atoms G* $\subseteq$ *atoms* (*Or F1 F2*) **using** *5 6* **by** (*rule subset-trans*)
next
case *False*
then have *G* $\notin$ *setSubformulae F1* **by** *simp*
then have *G* $\in$ *setSubformulae F2* **using** *4* **by** *simp*
then have *7*:*atoms G* $\subseteq$ *atoms F2* **using** *H5* **by** *simp*
have *8*:*atoms F2* $\subseteq$ *atoms* (*Or F1 F2*) **by** *simp*
show *atoms G* $\subseteq$ *atoms* (*Or F1 F2*) **using** *7 8* **by** (*rule subset-trans*)
qed
qed
next
case (*Imp F1 F2*)
then show *?case* **by** *auto*
qed

Por último, su demostración aplicativa y automática con la definición no simplificada.

lemma *subformula-atoms-aplicativa*: *G* $\in$ *setSubformulae F* $\Longrightarrow$ *atoms G* $\subseteq$ *atoms F*
apply (*induction F*)
apply *auto*
done

lemma *subformula-atoms*: *G* $\in$ *setSubformulae F* $\Longrightarrow$ *atoms G* $\subseteq$ *atoms F*
by (*induction F*) *auto*

A continuación voy a introducir un lema que no pertenece a la teoría original de Isabelle pero facilita las siguientes demostraciones detalladas mediante contenciones en cadena.

Lema 1.6 *Sea* $G \in SubfSet(F)$*, entonces* $SubfSet(G) \subseteq SubSet(F)$.

Veamos sus demostraciones según las distintas tácticas.

lemma *subsubformulae-estructurada*:
G $\in$ *setSubformulae F* $\Longrightarrow$ *setSubformulae G* $\subseteq$ *setSubformulae F*
proof (*induction F*)
case (*Atom x*)
then show *?case* **by** *simp*

next
  case *Bot*
  then show *?case* **by** *simp*
next
  case (*Not F*)
  assume *H1*:$G \in$ *setSubformulae F* $\Longrightarrow$ *setSubformulae G* $\subseteq$ *setSubformulae F*
  assume *H2*:$G \in$ *setSubformulae* (*Not F*)
  then show *setSubformulae G* $\subseteq$ *setSubformulae* (*Not F*)
  proof (*cases G* = *Not F*)
    case *True*
    then show *?thesis* **by** *simp*
  next
    case *False*
    then have $G \neq$ *Not F* **by** *simp*
    then have $G \in$ *setSubformulae F* **using** *H2* **by** *simp*
    then have *1*:*setSubformulae G* $\subseteq$ *setSubformulae F* **using** *H1* **by** *simp*
    have *setSubformulae* (*Not F*) = {*Not F*} $\cup$ *setSubformulae F* **by** *simp*
    have *setSubformulae F* $\subseteq$ {*Not F*} $\cup$ *setSubformulae F* **by** (*rule Un-upper2*)
    then have *2*:*setSubformulae F* $\subseteq$ *setSubformulae* (*Not F*) **by** *simp*
    show *setSubformulae G* $\subseteq$ *setSubformulae* (*Not F*) **using** *1 2* **by** (*rule subset-trans*)
  qed
next
  case (*And F1 F2*)
  then show *?case* **by** *auto*
next
  case (*Or F1 F2*)
  then show *?case* **by** *auto*
next
  case (*Imp F1 F2*)
  assume *H3*:$G \in$ *setSubformulae F1* $\Longrightarrow$ *setSubformulae G* $\subseteq$ *setSubformulae F1*
  assume *H4*:$G \in$ *setSubformulae F2* $\Longrightarrow$ *setSubformulae G* $\subseteq$ *setSubformulae F2*
  assume *H5*:$G \in$ *setSubformulae* (*Imp F1 F2*)
  have *4*:*setSubformulae* (*Imp F1 F2*) = {*Imp F1 F2*} $\cup$ (*setSubformulae F1* $\cup$ *setSubformulae F2*)
    by *simp*
  then show *setSubformulae G* $\subseteq$ *setSubformulae* (*Imp F1 F2*)
  proof (*cases G* = *Imp F1 F2*)

case *True*
then show *?thesis* **by** *simp*
next
case *False*
then have *5*:$G \neq$ *Imp F1 F2* **by** *simp*
have *setSubformulae F1* $\cup$ *setSubformulae F2* $\subseteq$
{*Imp F1 F2*} $\cup$ (*setSubformulae F1* $\cup$ *setSubformulae F2*) **by** (*rule Un-upper2*)
then have *6*:*setSubformulae F1* $\cup$ *setSubformulae F2* $\subseteq$ *setSubformulae* (*Imp F1 F2*) **by** *simp*
then show *setSubformulae G* $\subseteq$ *setSubformulae* (*Imp F1 F2*)
proof (*cases G* $\in$ *setSubformulae F1*)
case *True*
then have *G* $\in$ *setSubformulae F1* **by** *simp*
then have *7*:*setSubformulae G* $\subseteq$ *setSubformulae F1* **using** *H3* **by** *simp*
have *8*:*setSubformulae F1* $\subseteq$ *setSubformulae* (*Imp F1 F2*) **using** *6* **by** (*rule subContUnionRev1*)
show *setSubformulae G* $\subseteq$ *setSubformulae* (*Imp F1 F2*) **using** *7 8* **by** (*rule subset-trans*)
next
case *False*
then have *9*:*G* $\notin$ *setSubformulae F1* **by** *simp*
have *G* $\in$ *setSubformulae F1* $\cup$ *setSubformulae F2* **using** *5 H5* **by** *simp*
then have *G* $\in$ *setSubformulae F2* **using** *9* **by** *simp*
then have *10*:*setSubformulae G* $\subseteq$ *setSubformulae F2* **using** *H4* **by** *simp*
have *11*:*setSubformulae F2* $\subseteq$ *setSubformulae* (*Imp F1 F2*) **using** *6* **by** *simp*
show *setSubformulae G* $\subseteq$ *setSubformulae* (*Imp F1 F2*) **using** *10 11* **by** (*rule subset-trans*)
qed
qed
qed

lemma *subformulae-setSubformulae*:*G* $\in$ *setSubformulae F* $\Longrightarrow$ *setSubformulae G* $\subseteq$ *setSubformulae F*
by (*induction F*) *auto*

El siguiente lema nos da la noción de transitividad de contención en cadena de las subfórmulas, de modo que la subfórmula de una subfórmula es del mismo modo subfórmula de la mayor. Es un resultado sencillo derivado de la estructura de árbol de formación que siguen las fórmulas según las hemos definido.

Lema 1.7 *Sea $G \in SubfSet(F)$ y $H \in SubfSet(G)$, entonces $H \in SubfSet(F)$.*

La demostración estructurada se hace de manera sencilla con el resultado introducido anteriormente. Veamos su prueba según las distintas tácticas.

lemma *subsubformulae-estruct*:
  assumes $G \in$ *setSubformulae F*
    $H \in$ *setSubformulae G*
  shows $H \in$ *setSubformulae F*
proof −
  have *1*:*setSubformulae G* $\subseteq$ *setSubformulae F* **using** *assms*(*1*) **by** (*rule subformulae-setSubformulae*)
  have *setSubformulae H* $\subseteq$ *setSubformulae G* **using** *assms*(*2*) **by** (*rule subformulae-setSubformulae*)
  then have *2*:*setSubformulae H* $\subseteq$ *setSubformulae F* **using** *1* **by** (*rule subset-trans*)
  have $H \in$ *setSubformulae H* **by** *simp*
  then show $H \in$ *setSubformulae F* **using** *2* **by** (*rule rev-subsetD*)
qed

lemma *subsubformulae-aplic*:
  $G \in$ *setSubformulae F* $\Longrightarrow$ $H \in$ *setSubformulae G* $\Longrightarrow$ $H \in$ *setSubformulae F*
  oops

lemma *subsubformulae*: $G \in$ *setSubformulae F* $\Longrightarrow$ $H \in$ *setSubformulae G* $\Longrightarrow$ $H \in$ *setSubformulae F*
  by (*induction F*; *force*)

A continuación presentamos otro resultado que relaciona los conjuntos de subfórmulas según las conectivas que operen.

lemma *subformulas-in-subformulas*:
  $G \wedge H \in$ *setSubformulae F* $\Longrightarrow$ $G \in$ *setSubformulae F* $\wedge$ $H \in$ *setSubformulae F*

$G \vee H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$

$G \rightarrow H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$

$\neg\ G \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F$

oops

Como podemos observar, el resultado es análogo en todas las conectivas binarias aunque aparezcan definidas por separado, por tanto haré la demostración estructurada para una de ellas pues el resto son equivalentes. Nos basaremos en el lema anterior *subsubformulae*.

lemma *subformulas-in-subformulas-conjuncion-estructurada*:
assumes $And\ G\ H \in setSubformulae\ F$
shows $G \in setSubformulae\ F \wedge H \in setSubformulae\ F$
proof (*rule conjI*)
have $1{:}setSubformulae\ (And\ G\ H) = \{And\ G\ H\} \cup setSubformulae\ G \cup setSubformulae\ H$ **by** *simp*
then have $2{:}G \in setSubformulae\ (And\ G\ H)$ **by** *simp*
have $3{:}H \in setSubformulae\ (And\ G\ H)$ **using** *1* **by** *simp*
show $G \in setSubformulae\ F$ **using** *assms 2* **by** (*rule subsubformulae*)
show $H \in setSubformulae\ F$ **using** *assms 3* **by** (*rule subsubformulae*)
qed

lemma *subformulas-in-subformulas-negacion-estructurada*:
assumes $Not\ G \in setSubformulae\ F$
shows $G \in setSubformulae\ F$
proof −
have $setSubformulae\ (Not\ G) = \{Not\ G\} \cup setSubformulae\ G$ **by** *simp*
then have $1{:}G \in setSubformulae\ (Not\ G)$ **by** *simp*
show $G \in setSubformulae\ F$ **using** *assms 1* **by** (*rule subsubformulae*)
qed

Mostremos ahora la demostración aplicativa y automática para el lema completo.

lemma *subformulas-in-subformulas-aplicativa-s*:
$And\ G\ H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$

$Or\ G\ H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$

$Imp\ G\ H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$

$Not\ G \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F$
apply ((*rule conjI*)+, (*erule subsubformulae*,*simp*)+)+
done

lemma *subformulas-in-subformulas*:
$G \wedge H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$
$G \vee H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$
$G \longrightarrow H \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F \wedge H \in setSubformulae\ F$
$\neg\ G \in setSubformulae\ F \Longrightarrow G \in setSubformulae\ F$
by (*fastforce elim*: *subsubformulae*)+

HASTA AQUÍ HE TRABAJADO: 24/11/19

Concluimos la sección de subfórmulas con un resultado que relaciona varias funciones sobre la longitud de la lista *subformulae F* de una fórmula *F* cualquiera.

lemma *length-subformulae*: *length* (*subformulae F*) = *size F*
oops

En prime lugar aparece la clase *size* de la teoría de números naturales .... Vamos a definir *size1* de manera idéntica a como aparace *size* en la teoría.

class *size1* =
fixes *size1* :: $'a \Rightarrow nat$

instantiation *nat* :: *size1*
begin

definition *size1-nat* **where** [*simp*, *code*]: *size1* (*n*::*nat*) = *n*

instance ..

end

Como podemos observar, se trata de una clase que actúa sobre un parámetro global de tipo *'a* cualquiera. Por otro lado, *instantation* define una clase de parámetros, en este caso los números naturales *nat* que devuelve como resultado. Incluye una definición concreta del operador *size1* sobre dichos parámetros. Además, el último *instance* abre una prueba que afirma

que los parámetros dados conforman la clase especificada en la definición. Esta prueba que nos afirma que está bien definida la clase aparece omitida utilizando .. .
En particular, sobre una fórmula nos devuelve el número de elementos de la lista cuyos elementos son los nodos y las hojas de su árbol de formación.

value $size(n::nat) = n$
value$size(5::nat) = 5$

Por otro lado, la función *length* de la teoría List.thy nos indica la longitud de una lista cualquiera de elementos, definiéndose utilizando el comando *size* visto anteriormente.

abbreviation $length :: 'a\ list \Rightarrow nat$ **where**
$length \equiv size$

Las demostración del resultado se vuelve a basar en la inducción que nos despliega seis casos. La prueba estructurada no resulta interesante, pues todos los casos son inmediatos por simplificación como en el primer lema de esta sección. Incluimos a continuación la prueba aplicativa y automática.

lemma *length-subformulae-aplicativa*: $length\ (subformulae\ F) = size\ F$
  apply (*induction F*)
  apply *simp-all*
 done

lemma *length-subformulae*: $length\ (subformulae\ F) = size\ F$
  by (*induction F*; *simp*)

1.2 Conectivas derivadas

En esta sección definiremos nuevas conectivas y elementos a partir de los ya definidos en el apartado anterior. Además veremos varios resultados sobre los mismos.

En primer lugar, vamos a definir $Top:: 'a\ formula \Rightarrow bool$ como la constante que devuelve el booleano contrario a *Bot*. Se trata, por tanto, de una constante de la misma naturaleza que la ya definida para *Bot*. De este modo, *Top* será equivalente a *Verdadero*, y *Bot* a *Falso*, según se muestra en la siguiente ecuación. Su símbolo queda igualmente retratado a continuación.

definition *Top* ($\top$) **where**
$\top \equiv \bot \to \bot$

Por la propia definición y naturaleza de *Top*, verifica que no contiene ninguna variable del alfabeto, como ya sabíamos análogamente para *Bot*. Tenemos así la siguiente propiedad.

lemma *top-atoms-simp*[*simp*]: *atoms* $\top = \{\}$
  unfolding *Top-def* **by** *simp*

A continuación vamos a definir dos conectivas que generalizarán la conjunción y la disyunción para una lista finita de fórmulas. .

primrec *BigAnd* :: *'a formula list* $\Rightarrow$ *'a formula* ($\bigwedge$-) **where**
  $\bigwedge Nil = (\neg\bot)$
| $\bigwedge(F\#Fs) = F \wedge \bigwedge Fs$

primrec *BigOr* :: *'a formula list* $\Rightarrow$ *'a formula* ($\bigvee$-) **where**
  $\bigvee Nil = \bot$
| $\bigvee(F\#Fs) = F \vee \bigvee Fs$

Ambas nuevas conectivas se caracterizarán por ser del tipo funciones primitivas recursivas. Por tanto, sus definiciones se basan en dos casos:

Lista vacía: Representada como *Nil*. En este caso, la conjunción plural aplicada a la lista vacía nos devuelve la negación de *Bot*, es decir, *Verdadero*, y la disyunción plural sobre la lista vacía nos da simplemente *Bot*, luego *Falso*.

Lista recursiva: En este caso actúa sobre $F\#Fs$ donde F es una fórmula concatenada a la lista de fórmulas *Fs*. Como es lógico, *BigAnd* nos devuelve la conjunción de todas las fórmulas de la lista y *BigOr* nos devuelve su disyunción.

Además, se observa en cada función el símbolo de notación que aparece entre paréntesis. La conjunción plural nos da el siguiente resultado.

lemma *atoms-BigAnd*[*simp*]: *atoms* $(\bigwedge Fs) = \bigcup(atoms\ \text{‘}\ set\ Fs)$
  by(*induction Fs*; *simp*)

Referencias

[1] José A. Alonso. Temas de "Lógica matemática y fundamentos (2018–19)". Technical report, Univ. de Sevilla, 2019. En https://www.cs.us.es/~jalonso/cursos/lmf-18/temas.php.

[2] M. Fitting. *First-order Logic and Automated Theorem Proving*. Graduate texts in computer science. Springer, 1996.

[3] Tobias Nipkow, Lawrence C. Paulson, and Markus Wenzel. *Isabelle/HOL: A proof assistant for Higher–Order Logic*. Lecture Notes in Computer Science, Vol. 2283, Springer–Verlag, 2019. En https://www.cl.cam.ac.uk/research/hvg/Isabelle/dist/Isabelle2019/doc/tutorial.pdf.